


# Linkage File for 2018 MEPS and 2016/2017 NHIS Public Use Files

## 1.0 Overview

The Medical Expenditure Panel Survey (MEPS) Household Component (HC) survey uses the National Health Interview Survey (NHIS) as its sampling frame. Each year a new MEPS-HC panel is established, drawing from the previous year's NHIS sample. The MEPS-HC collects data through an overlapping panel design. In this design, two calendar years of information are collected from each household through in-person interviews, conducted over five rounds, with Round 3 spanning both calendar years.

The MEPS full-year public use files (PUFs) cover the calendar year of the data year and contain data from Rounds 3, 4, and 5 of the MEPS first panel (which uses the NHIS prior data year 1 as its sampling frame) combined with data from Rounds 1, 2, and 3 of the MEPS second panel (which uses the NHIS prior data year 2 as its sampling frame). As illustrated below, for full calendar year 2018 estimates, Rounds 3, 4, and 5 of Panel 22 (which uses the 2016 NHIS as its sampling frame) are combined with Rounds 1, 2, and 3 of Panel 23 (which uses the 2017 NHIS as its sampling frame).



PUFs containing NHIS data for a given calendar year are available from the National Center for Health Statistics (NCHS).

Users who need to augment the MEPS data with information from NHIS can do so with the linkage file described in the following sections.

## 2.0 Linkage File Description

The MEPS and NHIS linkage file, NHMEP18X.DAT, allows the data user to merge any of the person-level MEPS 2018 full-year public use data files with the 2016 and 2017 NHIS person-level public use data files (Person, Sample Adult, and Sample Child).

The NHIS person identifiers have been changed since 2004. Prior to 2004, one household equaled one "case," and unique person identifiers were Household Serial Number (HHX) and Person Number (PX). Since 2004, each family (FMX) has been considered a separate case, and unique person identifiers have been Household Serial Number (HHX), Family Sequence Number (FMX), and Person Sequence Number (FPX).

The linkage file contains 30,461 person-level records and seven variables. In the linkage file, a record

exists for each of the MEPS 2018 full-year persons. Each record contains the MEPS sample person ID (DUPERSID) and the corresponding NHIS sample person ID (HHX, FMX, and FPX). The linkage file can be linked to any of the person-level MEPS 2018 full-year public use data files using the variable DUPERSID. The linkage file can be linked to the NHIS 2016 or 2017 person-level data files by HHX, FMX, FPX, and SRVY\_YR.

When a MEPS sample person does not link to NHIS, HHX is set to 999999, FMX is set to 99, PX is set to 99, FPX is set to 99, SRVY\_YR is set to 9999, and LINKFLAG is set to 0.

### 3.0 Linkage File Record Counts

Of the 14,988 MEPS Panel 22 persons, 13,589 persons link to the 2016 NHIS data, while 14,396 of the 15,473 Panel 23 persons link to the 2017 NHIS data. A total of 2,476 persons in the two panels do not link to either 2016 or 2017 NHIS data. These unlinked cases include newborns; newly in-scope persons; and a small number of cases where the NHIS identified a household as responding, but when fielded in MEPS it was determined to actually be a nonresponding household. The table below summarizes the linkages.

**Table 1. Linkage file record counts**

2018 MEPS Full-Year Data	Linked to 2016 NHIS PUF (n = 97,169)	Linked to 2017 NHIS PUF (n = 78,132)	Linked to 2016 or 2017 NHIS PUF (n=175,301)	Not Linked to NHIS	Total
Panel 22 persons	13,589	0	13,589	1,399	14,988
Panel 23 persons	0	14,396	14,396	1,077	15,473
<b>Total</b>	13,589	14,396	27,985	2,476	30,461

### 4.0 Linkage File Record Layout

Following is the record layout for the person-level MEPS/NHIS linkage file (NHMEP18X.DAT).

Variable	Column position	Type	Label and value range
<b>DUPERSID</b>	1–10	Character	MEPS encrypted person ID (range = 10001101–99692104)
<b>HHX</b>	11–16	Character	NHIS household serial number (range = 000011–064978)
<b>FMX</b>	17–18	Character	NHIS family number (range=01–05)
<b>FPX</b>	19–20	Character	NHIS person number (range = 01–12)
<b>LINKFLAG</b>	21–21	Numeric	Linkage status between MEPS and NHIS (1 or 0)
<b>PANEL</b>	22–23	Numeric	MEPS panel number (either 22 or 23)
<b>SRVY_YR</b>	24–27	Numeric	NHIS survey year (2016 or 2017)

Below is the input statement to convert the linkage file (NHMEP18X.DAT) to a SAS dataset.

```
DATA XX.NHMEP18X;
  INFILE "C:\TEMP\MEPS\NHMEP18X.DAT";
  INPUT DUPERSID $1-10 HHX $11-16 FMX $17-18 FPX $19-20 LINKFLAG 21 PANEL 22-23 SRVY_YR 24-27;
RUN;
```

## 5.0 Linking Instructions for SAS Users

The following is one way of adding NHIS person-level variables to the MEPS person-level file. Input files are MEPS HC-204 (2018 Full-Year Population Characteristics), the 2016 NHIS person-level data file, the 2017 NHIS person-level data file, and the linkage file NHMEP18X.DAT.

- (1) Create four SAS datasets as follows:
  - Convert MEPS HC-204 (ASCII or SAS transport file) to a SAS dataset named FY2018 (n = 30,461).
  - Convert the linkage file NHMEP18X.DAT to a SAS dataset named NHMEP18X (n = 30,461).
  - Convert the 2016 NHIS Person file to a SAS dataset named NHIS2016 (n = 97,169). Make sure the SAS dataset includes HHX, FMX, FPX, SRVY\_YR, and other variables that are to be added to the MEPS full-year dataset.
  - Convert the 2017 NHIS Person file to a SAS dataset named NHIS2017 (n = 78,132). Make sure the SAS dataset includes HHX, FMX, FPX, SRVY\_YR, and other variables that are to be added to the MEPS full-year dataset.
- (2) Sort FY2018 by DUPERSID. Concatenate NHIS2016 and NHIS2017 into one dataset named NHISALL (n = 175,301). Sort NHISALL by HHX, FMX, FPX, and SRVY\_YR.
- (3) Merge FY2018 (n = 30,461) with NHMEP18X (n = 30,461) by DUPERSID. Name the output dataset MEPS (n = 30,461). Then sort MEPS by HHX, FMX, FPX, and SRVY\_YR.
- (4) Merge MEPS (n = 30,461) with NHISALL (n = 175,301) by HHX, FMX, FPX, and SRVY\_YR. Keep records only in MEPS (n = 30,461). Name the output dataset MEPS18NH (n = 30,461).

**Below is a sample SAS program for adding NHIS variables to the MEPS dataset.**

```
LIBNAME MEPS "C:\TEMP\MEPS"; /*MEPS 2018 Full-Year PUF, MEPS-NHIS Link, output file*/
LIBNAME NHIS "C:\TEMP\NHIS"; /*NHIS 2016 and 2017 Person Files*/

PROC FORMAT;
  VALUE AGE
    .='.'
    0-HIGH='>=0';
RUN;
```

```

DATA NHISALL;
  SET NHIS.NHIS2016 (KEEP=HHX FMX FPX SRVY_YR AGE_P /*other NHIS variables*/)
    NHIS.NHIS2017 (KEEP=HHX FMX FPX SRVY_YR AGE_P /*other NHIS variables*/);
RUN;

PROC SORT DATA=NHISALL;
  BY HHX FMX FPX SRVY_YR;
RUN;

DATA MEPS;
  MERGE MEPS.FY2018 MEPS.NHMEP18X (KEEP=DUPERSID HHX FMX FPX SRVY_YR LINKFLAG);
  BY DUPERSID;
RUN;

PROC SORT DATA=MEPS;
  BY HHX FMX FPX SRVY_YR;
RUN;

DATA MEPS.MEPS18NH;
  MERGE MEPS (IN=A) NHISALL;
  BY HHX FMX FPX SRVY_YR;
  IF A;
RUN;

TITLE1 "MEPS 2018 FY data with NHIS variables";
PROC FREQ DATA=MEPS.MEPS18NH;
  TABLES LINKFLAG*SRVY_YR*AGE_P/LIST MISSING;
  FORMAT AGE_P AGE.;
RUN;

```

## 6.0 Sample Stata Code for Adding NHIS Variables to the MEPS Dataset

```

log using stata18.log, replace

local meps c:\temp\meps
local nhis c:\temp\nhis

use `meps'/h204, clear
sort dupersid
save fy2018, replace

use `nhis'\nhis2016, clear
append using `nhis'\nhis2017
sort hhx fmx fpx srvy_yr
save nhisall, replace

use `meps'\nhmep18x, clear
sort dupersid
save link, replace

use fy2018
merge 1:1 dupersid using link
drop _merge
sort hhx fmx fpx srvy_yr
save meps, replace

merge m:1 hhx fmx fpx srvy_yr using nhisall
keep if _merge != 2
keep dupersid hhx fmx fpx srvy_yr linkflag /*plus nhis variables wanted to add*/

```

save `meps'/meps18nh, replace

describe

tab srvy\_yr linkflag, missing

## **7.0 Further Information**

For any questions regarding the linkage file, please contact May Chu at 301–427–1653 or by email at [May.Chu@ahrq.hhs.gov](mailto:May.Chu@ahrq.hhs.gov). MEPS public use data files can be downloaded free of charge from the MEPS website at <https://www.meps.ahrq.gov>. NHIS public use data files can be obtained by contacting NCHS by telephone (301–458–4636) or through their website, <https://www.cdc.gov/nchs>.